
概要

マイクで拾った音の大きさを5ドットのLEDで表示します。
コンデンサマイクの出力をトランジスタ1石で検波してサウンドレベル

に応じた直流電圧に変換し、IchigoJamのアナログ入力に入力します。

組み立てかた

SUL-jamP

※電解コンデンサとLEDは足の
長いほうがプラス側です

エミッタ(E)

コレクタ(C)

ベース(B)

部品表 ※予告なく変更することがあります

回路図 ※予告なく変更することがあります

5ドットLED(赤)

抵抗 270Ω
(赤紫茶金)

x3

抵抗 270Ω
(赤紫茶金) x2

抵抗 1MΩ
(茶黒緑金)

抵抗 3kΩ
(橙黒赤金)

積層セラミック
コンデンサ 0.1μF

(104)

トランジスタ
2SC1740

積層セラミック

コンデンサ 0.1μF
(104)

抵抗 13kΩ

(茶橙橙金)

抵抗 180Ω
(茶灰茶金)

電解コンデンサ
100μF(極性注意)

コンデンサ
マイク

トランジスタ
のピン配置

IchigoJamとの接続

1

2

※ジャンプワイヤの色は配線の区別のため色分け
してあります。(何色の線で配線してもかまいません)

100μF

品名/型番/値 数量 備考 品名/型番/値 数量 備考

1 5ドットLED 赤 1 7 抵抗 13kΩ 1 茶橙橙金の色帯

2 トランジスタ 2SC1740 1 8 抵抗 1MΩ 1 茶黒緑金の色帯

3 コンデンサマイク 1 9 積層セラミックコンデンサ 0.1μF 2 104の表記

4 抵抗 180Ω 1 茶灰茶金の色帯 10 電解コンデンサ 100μF 1

5 抵抗 270Ω 5 赤紫茶金の色帯 11 ジャンプワイヤー 長さ10cm 12

6 抵抗 3kΩ 1 橙黒赤金の色帯 12 ブレッドボード 1

IchigoJam 0.9.7/1.0.1対応

2015年 7月 SUL-jamP_v10_150713

本キットのプログラムはIchigoJam バージョン1.0.1用です。
プログラムの一部変更でバージョン0.9.7でも実行できます。

AM11:00～PM8:00



2 はじめのプログラム

プログラムの説明

アナログ電圧の取り込み
IchigoJamにはアナログ電圧を取り込む機能(ANA(2)関数)があり

ます。(ANA(2)関数については「キーワードの説明」を参照して
ください)
プログラムをRUNしてマイクに向かって話しかけると、一定時間

(約167ミリ秒)ごとにサウンドレベルに応じたアナログ電圧を取り込み、
モニタ画面に表示します。

トランジスタ1石で信号レベルをはかる

本キットの回路では、コンデンサ
マイクの音声信号出力をIchigoJam
で扱えるレベルにするのに、トラン

ジスタ1石による検波回路を使用
しています。

マイクから入力信号が入ると、コレクタ電流の平均値が変化します。

出力電圧が信号レベルに応じて変化するため、信号レベルをはかる
ことができます。

部品と回路の説明

サウンドレベルを一定時間ごとに取り込み、モニタ画面にその値を
表示するプログラムです。
マイクに向かって話しかけると、モニタ画面に表示される値が変化

します。

プログラムリスト

1 REM ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ VER.1.0
2 REM 2015/07/13
10 PRINT "ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ"
20 V=ANA(2)
30 PRINT V
40 WAIT 10
50 GOTO 20

キーワード
変数 V

ANA(2)関数 WAIT文

SAVE0

プログラムを書いたあと「SAVE0 」で
プログラムを保存してください。
( は[ENTER]キーです)
※0番に大事なプログラムが入っている
場合は、他の番号を指定して保存して
ください。

40 WAIT 10
50 GOTO 20

REM文(コメント)

SAVEコマンドの使い方

SAVE0

130 GOTO 30

SAVEコマンドは、作成したプログラム
を保存するコマンドです。

プログラムをSAVEしないと、Ichigo
Jamの電源をOFFにしたときに作成した
プログラムが消えてしまいます。

SAVEコマンドの後ろにプログラム番号(IchigoJam本体に
保存するときは0～3、EEPROMカセットに保存するときは
100～131)を指定して保存します。

SAVEすると指定した番号に入っていたプログラムは
上書きされます(消えてなくなります)ので十分注意して
ください。

SAVEコマンドで保存したプログラム
は、LOADコマンドでプログラム番号を
指定すると呼び出すことができます。

LOAD0
Loaded xxxbyte
OK

キーワードの説明

REM文(コメント) (1行)

REM文は、プログラムにコメント(メモ)を入れるために使用します。

(例) REM ｽｲｯﾁ ｦ ｵｽﾏﾃﾞ ﾏﾂ

「REM」の後ろはプログラムの実行時は無視されます。

変数 V (20行)

変数とは、プログラム中で使用する数値を入れるための「箱」です。

箱を区別するために、箱には「名前」をつけます。(これを「変数名」
といいます)

数値

コピー(代入)

箱に式の計算結果や別の箱の中身を入れる
(コピーする)ことを「代入」といいます。
(※コピー元の箱の中身は変化しません)

IchigoJamのBASICでは、変数名は「A」から「Z」までのアルファ
ベット1文字ですので、全部で26個の変数が使用できます。

(参考 :BASICでは変数を使用する前に「この名前の変数を

使用します」と宣言する必要はありません)

(参考2 :IchigoJamのBASICで扱える数値の範囲は、-32768から
32767までの整数です)

N

「N」という名前の変数に数値を代入するには、次のように書きます。

(例) N = 5

(例) N = K

※コピー元の変数(K)の値は変化しません

足し算や引き算などの計算にも使用します。

(例) N = J + K

N J K

2 3

N

5

IchigoJam バージョン0.9.7の場合
IchigoJam バージョン0.9.7を使用する場合は、プログラム

の20行を次のように変更してください。

20 V=ANA() 赤の下線部分が
変更箇所です

プログラムを1行だけ変更したいときは、変更したい行番号
で文を新しく書き、最後に[ENTER( )]キーを押すと新しく
書いた文で置き換わります。



WAIT文 (40行)

(例) WAIT 60

この例では、プログラムの実行を1秒(1/60秒 x 60 = 1秒)だけ
一時停止させます。

(例) WAIT 29-M

この例のように変数を含んだ式を指定すると、待ち時間を可変に
することも可能です。

一時停止させる時間として、定数の代わりに変数を含んだ式を
指定することもできます。

WAIT文は、プログラムの実行を指定した時間だけ一時停止させる
文です。一時停止させる時間は1/60秒(約16.7ミリ秒)単位で指定
します。

ANA(2)関数 (20行)

ANA(2)関数は、IchigoJamのアナログ入力端子(IN2端子)とGND
端子間のアナログ電圧を0～1023の整数で返します。

IN2の
電圧(V)

3.300 1023

3.296 1022

3.294 1021

0.006 2

0.003 1

0.000 0

ANA(2)の値

(例) N=ANA(2)

この例は、IN2端子のアナログ電圧に
対応した値を変数Nに代入しています。

IN2端子の電圧とANA(2)の値の関係は

左の表のようになっていますので、
ANA(2)の値を調べることにより、IN2端子
の電圧を知ることができます。

「ANA(2)」という名前の箱の中にIN2端子とGND端子間のアナログ
電圧を0～1023の整数で表したものがいつも入っています。
箱の中身はプログラム中でいつでも取り出すことができます。

アナログ電圧
(0～1023)

3 入門プログラム
はじめのプログラムを改良して、サウンドレベルを5ドットのLEDで

表示するようにします。
サウンドレベルを取り込み、そのレベルに応じて5ドットのLEDを

棒グラフ状に点灯させます。

プログラムリスト

1 REM ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ VER.1.0
2 REM 2015/07/13
10 PRINT "ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ"

20 REM ｼｮｷﾁ ﾉ ｾｯﾃｲ
30 K=640
40 REM ﾒｲﾝ ﾙｰﾌﾟ
50 V=ANA(2)

60 W=V-K
70 PRINT W
80 IF W>=10 THEN OUT 1,1 ELSE OUT 1,0

90 IF W>=20 THEN OUT 2,1 ELSE OUT 2,0
100 IF W>=30 THEN OUT 3,1 ELSE OUT 3,0
110 IF W>=40 THEN OUT 4,1 ELSE OUT 4,0

120 IF W>=50 THEN OUT 5,1 ELSE OUT 5,0
130 WAIT 5
140 GOTO 50

キーワード

Kの値(補正値)の設定方法
については、「プログラムの
説明」を見てください。

IF文と条件判断

プログラムの説明

補正値(K)の値の決め方(30行)

本キットに使用している部品(特にトランジスタ)の特性のばらつき
により、画面に表示されるサウンドレベルの値がずっとマイナスの
ままの場合や、大きすぎる場合があります。

このような場合はLEDの点灯がうまくいきませんので、プログラム
30行のKの値を変更する必要があります。

PRINT ANA(2)
635
OK

30 K=635

プログラム30行で設定する
Kの値を求めるには、マイクに
なるべく音が入らないようにして、
「PRINT ANA(2)」と入力します。
(行番号は不要です)

画面に補正値を表す数字が
表示されますので、プログラム
30行の数字を表示された数字に
変更してください。

はじめのプログラムで実験したように、マイクに向かって話しかける
と、サウンドレベルに応じてアナログ入力の値が無信号時の値から
増えます。

そこで、このプログラムでは音がしていないときのアナログ入力の
値を補正値としてあらかじめ用意しておき、一定時間ごとにはかった
アナログ入力の電圧との差をサウンドレベルとして扱っています。

プログラムをRUNしてマイクに向かって話しかけると、サウンド
レベルを5ドットのLEDで棒グラフ風に表示します。

参考(BASICの文の即時実行)

行番号なしでBASICの文を入力し、[ENTER]キーを押すと、
入力した文を直ちに実行することができます。

IchigoJam バージョン1.0.1では、かっこ内の数字(2)を省略
するとBTN端子のアナログ電圧を取り込みます。
IN2端子のアナログ電圧を取り込むにはANA(2)と書く必要
があります。

IchigoJam バージョン0.9.7のプログラムでANA()関数を
使用しているプログラムをIchigoJam バージョン1.0.1で動作
させるには、次のように書き換えてください。

<バージョン0.9.7><バージョン1.0.1>
(例) V=ANA()V=ANA(2)

※IchigoJam バージョン0.9.7を使用する場合は、プログラムの
50行を次のように変更してください。

50 V=ANA() 変更箇所



キーワードの説明

IF文と条件判断 (80行)

IF文は、条件によって処理内容を変えたいときに使用します。

(例) IF V>=186 THEN OUT 1,1

条件式 条件式が真のとき
実行する文

この例では、Vの値が186以上かどうか比較し、もし真ならば
(以上ならば)「THEN」以下の文を実行します。(偽ならば(未満ならば)
次の行に移ります)

条件式が偽のときに実行する文を書く必要があるときは、条件式が
真のときに実行する文に続けて「ELSE」と書き、その後ろに条件式が
偽のときに実行する文を書きます。

(例) IF V>=186 THEN OUT 1,1 ELSE OUT 1,0

条件式 条件式が真のとき
実行する文

条件式が偽のとき
実行する文

この例ではVの値が186以上かどうか比較し、もし真ならば
(V>=186ならば)OUT1端子に「1」を、偽ならば(V<186ならば)OUT1
端子に「0」を出力します。

プログラムの説明

(1) 補正値の自動設定(20～70行)

入門プログラムでは、補正値を手動設定で修正していました。
この補正値の設定を自動化することを考えます。

はじめのプログラムでの「キーワードの説明」のところで、ANA(2)
関数の使い方を説明しました。入門プログラムの最初の部分で
補正値(K)の値にANA()関数の値を代入すれば良さそうです。

しかし、この方法ではうまくいきません。アナログ信号でも温度や
明るさなど、変化の少ない信号ならばこの方法でもうまく補正値を
設定できますが、サウンドレベルのように激しく変化する信号には
適していません。

20 REM ｼｮｷﾁ ﾉ ｾｯﾃｲ
30 I=16:K=0:PRINT "ｼｮｷｶ ｼﾃｲﾏｽ"
40 K=K+ANA(2)
50 WAIT 10
60 I=I-1:IF I>0 THEN GOTO 40
70 K=K/16 (プログラムから抜粋)

そこで、一定時間ごとに
複数回サンプルをとり、
その平均値を補正値と
します。

左のリストでは16回
サンプルを取り、平均値を
求めています。

このプログラムではIchigoJam上のタクトスイッチを押すと
補正値の再設定が行えるようになっています。

上級プログラム

入門プログラムを改良して、補正値の初期設定を自動化します。
また、LEDの表示を実際の音量の感じに合わせてみます。

プログラムリスト

キーワード
コロン(「:」)記号 ループ構造

4

BTN()関数

(2) レベルの表示方法

このプログラムでは、サウンドレベルを5ドットのLEDで棒グラフ風

に表示しています。

120 IF W>=10 THEN OUT 1,1 ELSE OUT 1,0
130 IF W>=20 THEN OUT 2,1 ELSE OUT 2,0

140 IF W>=40 THEN OUT 3,1 ELSE OUT 3,0
150 IF W>=80 THEN OUT 4,1 ELSE OUT 4,0
160 IF W>=160 THEN OUT 5,1 ELSE OUT 5,0

(プログラムから抜粋)

人間の聴覚には、サウンドレベルと音量の感じ方との関係が
直線的でなく、サウンドレベルの対数に比例した音量の感じ方を
するという特徴があります。

このため、音声や音楽の信号を扱う電子回路では、「デシベル(dB)」
という単位をよく使います。(対数については高校の数学で習います。
デシベルについて興味のある方は電子回路の本を参照してください)

IF文で比較している値が10、20、40......というふうに2倍刻みになって
いる点に注目してください。これはLEDの点灯具合を実際の音量の
感じに合わせる工夫です。

このプログラムではIF文で比較する値を2倍刻みにすることで、
LEDの点灯具合が実際の音量の感じに近くなるようにしています。

※IchigoJam バージョン0.9.7を使用する場合は、プログラムの
40行と90行を次のように変更してください。

40 K=K+ANA() 変更箇所

90 V=ANA() 変更箇所

1 REM ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ VER.1.0
2 REM 2015/07/13
10 PRINT "ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ"
20 REM ｼｮｷﾁ ﾉ ｾｯﾃｲ
30 I=16:K=0:PRINT "ｼｮｷｶ ｼﾃｲﾏｽ"
40 K=K+ANA(2)
50 WAIT 10
60 I=I-1:IF I>0 THEN GOTO 40
70 K=K/16
80 REM ﾒｲﾝ ﾙｰﾌﾟ
90 V=ANA(2)
100 W=V-K
110 PRINT W
120 IF W>=10 THEN OUT 1,1 ELSE OUT 1,0
130 IF W>=20 THEN OUT 2,1 ELSE OUT 2,0
140 IF W>=40 THEN OUT 3,1 ELSE OUT 3,0
150 IF W>=80 THEN OUT 4,1 ELSE OUT 4,0
160 IF W>=160 THEN OUT 5,1 ELSE OUT 5,0

170 WAIT 5
180 IF BTN()=1 THEN GOTO 30 ELSE GOTO 90



キーワードの説明

コロン(「:」)記号(30行)

ループ構造 (30行～60行)

BASICのプログラムでは、1行に複数の文を書くことができます。
1行に複数の文を書くときは、文と文の間をコロン(「:」)記号で
区切って書きます。

(例) V=ANA():PRINT V

BTN()関数 (180行)

BTN()関数は、IchigoJam本体のタクトスイッチが押されていれば
「1」、押されていなければ「0」を返す関数です。

(例) IF BTN()=1 THEN GOTO 50

この例ではタクトスイッチを押した場合に50行にジャンプします。

上級プログラム・改

上級プログラムを改良して、LEDの点灯具合でVUメータの針の
振れ具合をまねてみます。

プログラムリスト

5

VUメータは左の写真のようなメータで、音声や音楽
の信号レベルを表示するのに使います。針の振れ
具合が実際の音量の感じに合うよう、針が少しゆっくり
振れるように工夫してあります。

プログラムの説明

表示の変化をゆっくりにするには
このプログラムでは、170行のWAIT文の待ち時間を「1」に指定

しています。(上級プログラムでは待ち時間を「5」に指定していました)
これは、拍手の音などの突発的な音もできるだけうまく拾えるように
と考えてのことです。

一定時間内になるべく多くの回数ANA(2)関数を呼び出すように
すると、突発的な音も拾えるチャンスが増えることになります。

WAIT 5のとき WAIT 1のとき

イメージ図

上図の○印のところでIN2端子のアナログ電圧がIchigoJamの
プログラムに取り込まれます。

このプログラムでは、サウンドレベルをきめ細かく取り込みながらも、

LEDの点灯具合をVUメータ風になめらかに変化するようにしています。
プログラムの105行がこのための計算です。

105 W=(W+X)/2 (プログラムより抜粋)

変数Xには、サウンドレベルの値が入っています。変数Wは記憶用
の変数で、フィルタした結果が入っています。

新しい値を取り込むごとに、変数Wの値とサウンドレベルの値Xとの
平均を取り(足して2で割る)、それを結果とします。結果はまた次回
値を取り込んだときに使いますので、変数Wに記憶しておきます。

この処理を行うと、表示の変化がなめらかになり、アナログ式(針式)
のVUメータ風になります。

IchigoJam バージョン1.0.1では、IchigoJam本体のタクト
スイッチの状態以外にキーボードの状態も取り込めます。
状態を調べたいキーは、BTN()関数のかっこの中に指定
します。

BTN(SPACE) :スペースキーの状態
BTN(UP) :上矢印キーの状態
BTN(DOWN) :下矢印キーの状態
BTN(LEFT) :左矢印キーの状態
BTN(RIGHT) :右矢印キーの状態

かっこの中身を省略した場合はIchigoJam本体のタクト
スイッチの状態を取り込みます。(バージョン0.9.7と同じ動作
をします)

※IchigoJam バージョン0.9.7を使用する場合は、プログラムの
40行と90行を次のように変更してください。

40 K=K+ANA() 変更箇所

90 V=ANA() 変更箇所

プログラム中で同じ処理を繰り返し行わせることがよくあります。
繰り返しのループ構造を書くには、次のように書きます。

(例)

80 M=64
90 BEEP 5,1
100 GOSUB 150
110 M=M-1
120 IF M>0 THEN GOTO 90

(1)残り回数セット

(2)ループ本体

(3)残り回数を減らす

(4) 残り回数が
0より大きければ
ループ本体を実行

(1) ループの残り回数を保持するための変数(M)にループを回る
回数(残り回数)をセットします。
(2) ループ本体の処理(繰り返し行わせる処理)を書きます。

(3) 残り回数を保持している変数(M)の値を1減らします。

(4) 残り回数を保持している変数(M)の値が0より大きいならば、

ループ本体の先頭にジャンプします。(Mの値が0の場合はジャンプ
せず、繰り返しループから抜けます)

IchigoJam バージョン1.0.1では、上に説明した方法以外
に、FOR/NEXT文を使った方法でもループ構造を作れます。

FOR/NEXT文がないIchigoJam バージョン0.9.7でも
同じプログラムで実験できるように、本キットではFOR/NEXT
文は使用していません。

1 REM ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ VER.1.0
2 REM 2015/07/13

10 PRINT "ｻｳﾝﾄﾞ ﾚﾍﾞﾙ ﾒｰﾀ"
20 REM ｼｮｷﾁ ﾉ ｾｯﾃｲ
25 W=0
30 I=16:K=0:PRINT "ｼｮｷｶ ｼﾃｲﾏｽ"
40 K=K+ANA(2)
50 WAIT 10
60 I=I-1:IF I>0 THEN GOTO 40
70 K=K/16
80 REM ﾒｲﾝ ﾙｰﾌﾟ
90 V=ANA(2)
100 X=V-K
105 W=(W+X)/2
110 PRINT W
120 IF W>=10 THEN OUT 1,1 ELSE OUT 1,0
130 IF W>=20 THEN OUT 2,1 ELSE OUT 2,0
140 IF W>=40 THEN OUT 3,1 ELSE OUT 3,0
150 IF W>=80 THEN OUT 4,1 ELSE OUT 4,0
160 IF W>=160 THEN OUT 5,1 ELSE OUT 5,0
170 WAIT 1
180 IF BTN()=1 THEN GOTO 25 ELSE GOTO 90

